EDGEIDEALS: A PACKAGE FOR (HYPER)GRAPHS

CHRISTOPHER A. FRANCISCO, ANDREW HOEFEL, AND ADAM VAN TUYL

ABSTRACT. We introduce a new package, entitled *EdgeIdeals*, that allows one to experiment with graphs and hypergraphs via the edge ideal correspondence. At the core of our package are new classes for defining graphs and hypergraphs.

1.1. **Introduction.** The edge ideal of a (hyper)graph enables one to study (hyper)graphs using the tools of commutative algebra. The work of [1, 3, 6, 7, 9, 11, 12, 13, 14, 15], among others, has focused on building a dictionary between commutative algebra and graph theory. In this short note, we wish to introduce a new package, entitled *EdgeIdeals*, that we have written for *Macaulay 2* [5], which exploits this dictionary. The goal of this package is to provide a family of functions that will enable the user to experiment with simple graphs and hypergraphs, thus facilitating future research. Many of the underlying algorithms in this package exploit the correspondence between (hyper)graphs and square-free monomial ideals, and as such, take advantage of previous *Macaulay 2* packages devoted to monomial ideals, e.g., the *SimplicialComplexes* package of Popescu, Smith, and Stillman (see [10] for more on monomial ideals and *Macaulay 2*).

1.2. **Mathematical Background.** A **hypergraph** is a pair $\mathcal{H} = (\mathcal{X}, \mathcal{E})$, where $\mathcal{X} = \{x_1, \ldots, x_n\}$ is the set of vertices, and $\mathcal{E} = \{E_1, \ldots, E_t\}$, a collection of subsets of $\mathcal{X}$, is the set of edges. A hypergraph is a **clutter** (or Sperner system) if the edges of $\mathcal{H}$ satisfy the property that whenever $E_i \subseteq E_j$, then $i = j$. If a simple hypergraph $\mathcal{H}$ has the property that $|E_i| = 2$ for each $E_i \in \mathcal{E}$, then we usually call $\mathcal{H}$ a **simple graph**.

The package *EdgeIdeals* is based upon the correspondence between simple graphs or clutters and square-free monomial ideals. Roughly speaking, the corresponding monomial ideal is used to find information about a given hypergraph, and conversely. The correspondence is defined as follows. Let $\mathcal{H} = (\mathcal{X}, \mathcal{E})$ be a clutter with vertex set $\mathcal{X} = \{x_1, \ldots, x_n\}$. Fix a field k and consider the polynomial ring $R = k[x_1, \ldots, x_n]$. We then identify each vertex of $\mathcal{H}$ with the corresponding variable in R. The edges of $\mathcal{H}$ are then used to define an ideal, called the **edge ideal** of $\mathcal{H}$, of the ring R:

$$I(\mathcal{H}) = \left\langle \left\{ \prod_{x \in E} x \;\middle|\; E \in \mathcal{E} \right\} \right\rangle.$$

Conversely, any square-free monomial ideal I minimally generated by $\{m_1, \ldots, m_s\}$ will correspond to a hypergraph where the edge E_i corresponds to the variables in the support of the monomial m_i. Note that what we call the edge ideal of a clutter

2000 *Mathematics Subject Classification.* 13F55, 05E99.
Key words and phrases. edge ideals, graphs, hypergraphs.

is equivalent to the notion of a **facet ideal** as first defined by Faridi [2] (see the discussion in [8]).

1.3. **Worked example with** ***EdgeIdeals*****.** At the heart of the *EdgeIdeals* package are two new classes: the first is a class entitled `HyperGraph`, and second is the class called `Graph`, which extends `HyperGraph` and inherits all of the methods of `HyperGraph`. Functions have been made that accept objects of either type as input.

In our example below, we will illustrate [15, Theorem 6.4.7] that says the independence complex of a Cohen-Macaulay bipartite graph has a simplicial shelling. We begin by inputting a graph and verifying the Cohen-Macaulay and bipartite properties.

```
i1 : loadPackage "EdgeIdeals";

i2 : R = QQ[x_1..x_3,y_1..y_3];

i3 : G = graph(R,{x_1*y_1,x_2*y_2,x_3*y_3, x_1*y_2,x_1*y_3,x_2*y_3})

o3 = Graph{edges => {{x , y }, {x , y }, {x , y }, {x , y }, {x , y }, {x , y }}}
                       1   1     2   2     3   3     1   2     1   3     2   3
           ring => R
           vertices => {x , x , x , y , y , y }
                         1   2   3   1   2   3

o3 : Graph

i4 : isCM G and isBipartite G

o4 = true
```

When defining a (hyper)graph, the user specifies the vertex set by defining a polynomial ring, while the edges are written as a list of square-free monomials (there are alternative ways of listing the edges). A (hyper)graph is stored as a hash table which contains the list of edges, the polynomial ring, and the list of vertices.

```
i5 : L = getGoodLeaf(G)

o5 = {x , y }
       1   1

o5 : List

i6 : degreeVertex(G,y_1)

o6 = 1

i7 : H = inducedHyperGraph(G, vertices(G) - set(L))

o7 = HyperGraph{edges => {{x , y }, {x , y }, {x , y }}}
                            2   2     3   3     2   3
                ring => QQ [x , x , y , y ]
                             2   3   2   3
                vertices => {x , x , y , y }
                              2   3   2   3

o7 : HyperGraph
```

A Cohen-Macaulay bipartite graph must contain a leaf, which we retrieve above. We remove the leaf, to form the induced graph, and at the same time, we identify the vertex of degree one in the leaf.

```
i8 : K = simplicialComplexToHyperGraph independenceComplex H;

i9 : edges K

o9 = {{x , x }, {x , y }, {y , y }}
        2   3     3   2     2   3

o9 : List
```

In the above, we formed the independence complex of H, that is, the simplicial complex whose facets correspond to the maximal independent sets of H. We then change the type from a simplicial complex to a hypergraph, which we call K. Notice that these edges give a shelling.

```
i10 : use ring K;
```

```
i11 : A = apply(edges(K), e->append(e, y_1));

i12 : B = apply(edges inducedHyperGraph(K, {x_2,x_3}), e-> append(e, x_1));

i13 : shelling = join(A,B)

o13 = {{x , x , y }, {x , y , y }, {y , y , y }, {x , x , x }}
         2   3   1     3   2   1     2   3   1     2   3   1

o13 : List

i14 : independenceComplex(G)

o14 = | y_1y_2y_3 x_3y_1y_2 x_2x_3y_1 x_1x_2x_3 |

o14 : SimplicialComplex
```

Using the method found in the proof of [15, Theorem 6.4.7], we now can form a shelling of the original independence complex. Notice that our shelling is a permutation of the facets of the independence complex defined from G.

References

[1] D. Eisenbud, M. Green, K. Hulek and S. Popescu, Restricting linear syzygies: algebra and geometry. *Compositio Math.* **141** (2005), 1460-1478.
[2] S. Faridi The facet ideal of a simplicial complex. Manuscripta Mathematica **109** (2002), 159–174.
[3] C.A. Francisco, H.T. Hà, Whiskers and sequentially Cohen-Macaulay graphs. *J. Combin. Theory Ser. A* **115** (2008), no. 2, 304–316.
[4] C.A. Francisco, H.T. Hà, A. Van Tuyl, Algebraic methods for detecting odd holes in a graph. (2008) Preprint. *arXiv:0806.1159v1*
[5] D. R. Grayson and M. E. Stillman, *Macaulay 2*, a software system for research in algebraic geometry. http://www.math.uiuc.edu/Macaulay2/.
[6] H.T. Hà, A. Van Tuyl, Resolutions of square-free monomial ideals via facet ideals: a survey. *Contemp. Math.* **448** (2007), 91-117.
[7] H.T. Hà, A. Van Tuyl, Splittable ideals and the resolutions of monomial ideals. *J. Algebra* **309** (2007), 405–425.
[8] H.T. Hà, A. Van Tuyl, Monomial ideals, edge ideals of hypergraphs, and their minimal graded free resolutions. *J. Algebraic Combin.* **27** (2008), no. 2, 215–245.
[9] J. Herzog, T. Hibi, and X. Zheng, Cohen-Macaulay chordal graphs. *J. Combin. Theory Ser. A* **113** (2006), 911–916.
[10] S. Hosten, G. Smith, Monomial ideals. In *Computations in algebraic geometry with Macaulay 2*, Algorithms and Computations in Mathematics **8** (2001), Springer-Verlag, 73–100, Springer-Verlag.
[11] S. Jacques, Betti numbers of graph ideals. Ph.D. Thesis, University of Sheffield, 2004. math.AC/0410107
[12] A. Simis, W.V. Vasconcelos, R.H. Villarreal, On the ideal theory of graphs. *J. Algebra* **167** (1994), 389-416.
[13] B. Sturmfels, S. Sullivant, Combinatorial secant varieties. *Pure Appl. Math. Q.* **2** (2006), 867–891.
[14] R.H. Villarreal, Cohen-Macaulay graphs. *Manuscripta Math.* **66** (1990), no. 3, 277–293.
[15] R.H. Villarreal, *Monomial algebras.* Monographs and Textbooks in Pure and Applied Mathematics, **238**. Marcel Dekker, Inc., New York, 2001.
[16] X. Zheng, Homological properties of monomial ideals associated to quasi-trees and lattices. PhD Thesis, Essen, (2004).

Department of Mathematics, Oklahoma State University, 401 Mathematical Sciences, Stillwater, OK 74078 chris@math.okstate.edu

Dalhousie University, Department of Mathematics & Statistics, Chase Building, Halifax, NS B3H 3J5, Canada andrew.hoefel@mathstat.dal.ca

Department of Mathematical Sciences, Lakehead University, Thunder Bay, ON P7B 5E1, Canada, avantuyl@sleet.lakeheadu.ca